

# Challenge November 2022

Crack a padlock

A solution with OPL CPLEX by Alex Fleischer  
afleischer@fr.ibm.com

OPL (Optimization Programming Language) is an algebraic modeling language that helps model easily optimization problems that can be solved both with IBM CPLEX linear programming and IBM CPLEX constraint programming CPOptimizer (CPO)

Optimization can help any kind of business, which includes for sure opening a padlock.


A padlock has a 4-digit key code.

2	6	5	7
---	---	---	---

Has two correct digits but neither are in the correct place.

4	2	6	8
---	---	---	---

Has no correct digits.



0	4	1	5
---	---	---	---

Has one correct digit but it's in the wrong place.

1	7	4	9
---	---	---	---

Has two correct digits, both in the correct place.

All the 4 digits in the key are different.  
What is the code for the padlock?

With OPL CPLEX CPOptimizer we can write a very simple model that could help you next time you have some time off.

NB: You can use free CPLEX Community Edition for this.

## The .mod (Model)

```
using CP;

range r=1..4;

// What we look for
dvar int x[r] in 0..9;
dvar int code;

tuple test
{
  int y[r];
  int nbPresent;
  int nbCorrectPlace;
}

{test} tests=...;

subject to
{
  allDifferent(x);

  forall(t in tests)
  {
    t.nbPresent==sum(i in r) or(j in r) (x[i]==t.y[j]);
    t.nbCorrectPlace==sum(i in r) (x[i]==t.y[i]);
  }

  code == 1000*x[1]+100*x[2]+10*x[3]+x[4];
}
```

And the .dat (data)

```
tests=
{<[2,6,5,7],2,0>,<[0,4,1,5],1,0>,
<[4,2,6,8],0,0>,<[1,7,4,9],2,2>};
```

Which gives

5739